Estimates for the Parabolic Renormalization

Hiroyuki Inou and Mitsuhiro Shishikura

About this file:

This file is a Maple work sheet to be used to check the numercal estimates in the proof of Main Theorem 1 of the paper:

Inou and Shishikura, The renormalization for parabolic fixed points and their perturbation.

Last updated 2006-0505.

5.A Outline

Definition of P(z) and Q(z), their relation

P(z) and Q(z). [Definition]

```
> P := z -> z*(1+z)^2;
  Q := z -> z*(1+1/z)^6/(1-1/z)^4;
```

$$P := z \to z\,(1+z)^2$$

$$Q := z \to \frac{z\left(1+\frac{1}{z}\right)^6}{\left(1-\frac{1}{z}\right)^4}$$

Functions which relate P and Q. [Definition]

```
> f_Koebe := z -> z/(1-z)^2;
  psi1 := z -> 4*f_Koebe(-1/z);
  psi0 := z -> -4/z; psi0_inverse := z -> -4/z;
```

$$f\_Koebe := z \to \frac{z}{(1-z)^2}$$

$$\psi 1 := z \to 4\, f\_Koebe\left(-\frac{1}{z}\right)$$

$$\psi 0 := z \to -\frac{4}{z}$$

$$psi0\_inverse := z \to -\frac{4}{z}$$

The relation between P and Q. This shows that Q(z) = phi0^{-1}(P(phi1(z))). [Formal Computation]

```
> psi0_inverse( P( psi1(z) ) );
  factor( simplify(%) ); simplify( Q(z) );
```

$$\frac{z\left(1+\frac{1}{z}\right)^2}{\left(1-\frac{4}{z\left(1+\frac{1}{z}\right)^2}\right)^2}$$

$$\frac{(z+1)^6}{z\,(z-1)^4}$$

$$\frac{(z+1)^6}{z\,(z-1)^4}$$

Ellipse E

Center, major and minor axes. [Definition]

```
> x_E := -0.18; a_E := 1.24; b_E := 1.04;
```

$$x\_E := -0.18$$

$$a\_E := 1.24$$

$$b\_E := 1.04$$

The unit disk and the ellipse. [plot]

```
> plot([[x_E+a_E*cos(t), b_E*sin(t),t=0..2*Pi],
           [cos(t), sin(t),t=0..2*Pi]],
           color=[red, black], scaling=constrained);
```



Shape of V

Boundary of V, which is the image of bdry E under psi1. [plot]

```
> with(plots):
  plot([Re(psi1(x_E+a_E*cos(t)+I*b_E*sin(t))),
          Im(psi1(x_E+a_E*cos(t)+I*b_E*sin(t))),t=0..2*Pi],
          color=blue, numpoints=300, scaling=constrained);
```

```
Warning, the name changecoords has been redefined
```

Blow up of V near -1. [Plot]

```
> replot(%, view=[-1.7..-0.9,-0.03..0.03], scaling=unconstrained);
  replot(%, view=[-1.005..-0.995,-0.005..0.005]);
```



0.03
0.02
0.01
0
-0.01
-0.02
-0.03
-1.6
-1.4
-1.2
-1



Eta, R, rho [Definition]

```
> eta := 2.0; R := 266; rho := 0.05;
```

$$\eta := 2.0$$

$$R := 266$$

$$\varrho := 0.05$$

5.B Preparation

Lemma 5.10

Conformal map for C-E. [Definition]

```
> e_1 := 1.14; e_0 := -0.18; e_minus1 := 0.1;
  zeta_E := w -> e_1*w + e_0 + e_minus1/w;
  a_Er := r -> e_1*r + e_minus1/r; b_Er := r -> e_1*r - e_minus1/r;
```

$$e\_1 := 1.14$$

$$e\_0 := -0.18$$

$$e\_minus1 := 0.1$$

$$zeta\_E := w \to e\_1\, w + e\_0 + \frac{e\_minus1}{w}$$

$$a\_Er := r \to e\_1\, r + \frac{e\_minus1}{r}$$

$$b\_Er := r \to e\_1\, r - \frac{e\_minus1}{r}$$

Lemma 5.11 (b)

Poincare disk and Euclidean disk. [Formal]
Let H={Re(z*exp(-I*theta)) > t} and z_0 = (t+u+I*v)*exp(I*theta).
A conformal map from H to D, sending z_0 to 0, is given by Map_H_to_D.

```
> Map_H_to_D := z -> (z - (t+u+I*v)*exp(I*theta))/(z -
  (t-u+I*v)*exp(I*theta));
  simplify(Map_H_to_D( (t+u+I*v)*exp(I*theta) +
  2*u*r^2*exp(I*theta)/(1-r^2) + 2*u*r*exp(I*theta)/(1-r^2) ));
  simplify(Map_H_to_D( (t+u+I*v)*exp(I*theta) +
  2*u*r^2*exp(I*theta)/(1-r^2) - 2*u*r*exp(I*theta)/(1-r^2) ));
```

$$Map\_H\_to\_D := z \to \frac{z - (t + u + I\,v)\,\mathbf{e}^{(I\,\theta)}}{z - (t - u + I\,v)\,\mathbf{e}^{(I\,\theta)}}$$

$$r$$

$$-r$$

5.C Covering property of f in F_0 and P as subcover

Domains and curves for P(z)

Parametrize gamma_bi, gamma_ci by t, so that P(Re_gamma_P(t) + I*Im_gamma_P(t)) = P(t).
[Definition and Formal]

```
> Re_gamma_P:= t -> -1-t/2; Im_gamma_P:= t -> sqrt(t*(4+3*t))/2;
  P( Re_gamma_P(t)+I*Im_gamma_P(t) ) - P(t); simplify(%);
```

$$Re\_gamma\_P := t \to -1 - \frac{1}{2}\,t$$

$$Im\_gamma\_P := t \to \frac{1}{2}\sqrt{t\,(4 + 3\,t)}$$

$$\left(-1 - \frac{1}{2}\,t + \frac{1}{2}\,I\sqrt{t\,(4 + 3\,t)}\right)\left(-\frac{1}{2}\,t + \frac{1}{2}\,I\sqrt{t\,(4 + 3\,t)}\right)^2 - t\,(1 + t)^2$$

$$0$$

gamma_ai (red), gamma_bi (blue), gamma_ci (green). [plot]

```
> t_1:= 2:
  plot([[x,0,x=-4/3..0],
           [x,0,x=(-4/3-t_1)..-4/3],
           [Re_gamma_P(t), Im_gamma_P(t), t=(-4/3-t_1)..-4/3],
           [Re_gamma_P(t), -Im_gamma_P(t), t=(-4/3-t_1)..-4/3],
           [x,0,x=0..t_1],
           [Re_gamma_P(t), Im_gamma_P(t), t=0..t_1],
           [Re_gamma_P(t), -Im_gamma_P(t), t=0..t_1]],
           color=[red, blue, blue, blue, green, green, green],
```

```
numpoints=100, scaling=constrained, axes=NONE);
```



5.D Passing from P to Q

Lemma 5.14.

(a) For Q(z)=psi0^{-1}(P(psi1(z))), see also 5.A. [Formal]

```
> simplify( P(psi1(z)) - psi0(Q(z)) );
  psi11 := z -> (z-1)/(z+1); psi12 := w -> w^2-1;
  simplify( psi12(psi11(z)) );
```

$$0$$

$$\psi 11 := z \rightarrow \frac{z-1}{1+z}$$

$$\psi 12 := w \rightarrow w^2 - 1$$

$$-\frac{4\,z}{(z+1)^2}$$

cv, cp, cp2, verfify the critical points. [Definition, Formal]

```
> cv := 27; cp := 5+2*sqrt(6); evalf(%); cp2 := 5-2*sqrt(6);
  evalf(%);
  simplify(D(Q)(cp)); simplify(D(Q)(cp2)); simplify(Q(cp));
```

```
simplify(Q(cp2));
```

$$cv := 27$$

$$cp := 5 + 2\sqrt{6}$$

$$9.898979486$$

$$cp2 := 5 - 2\sqrt{6}$$

$$0.101020514$$

$$0$$

$$0$$

$$27$$

$$27$$

Domains and curves for Q(z).

Define inverse branches of psi1. [Definition and Formal]

```
> Invpsi1 := w -> -(1+2/w)-(2/w)*sqrt(1+w);
  Invpsi1_plus := w -> -(1+2/w)+(2/w)*sqrt(1+w);
  simplify( psi1(Invpsi1(w)) - w ); simplify( psi1(Invpsi1_plus(w))
  - w );
```

$$Invpsi1 := w \to -1 - \frac{2}{w} - \frac{2\sqrt{1+w}}{w}$$

$$Invpsi1\_plus := w \to -1 - \frac{2}{w} + \frac{2\sqrt{1+w}}{w}$$

$$0$$

$$0$$

gamma_ai (red), gamma_bi (blue), gamma_ci (green), D(I/sqrt(3), 2/sqrt(3)) (yellow). [plot]

```
> t_1:= 5; t_2:= 1000;
  Invpsi1 := w -> -(1+2/w)-(2/w)*sqrt(1+w);
  Invpsi1_plus := w -> -(1+2/w)+(2/w)*sqrt(1+w);
  Re_gamma_Q:= t -> Re(Invpsi1(Re_gamma_P(t)+I*Im_gamma_P(t)));
  Im_gamma_Q:= t -> Im(Invpsi1(Re_gamma_P(t)+I*Im_gamma_P(t)));
  plot([         [(2/sqrt(3))*cos(t), 1/sqrt(3)+(2/sqrt(3))*sin(t),
  t=0..2*Pi],
           [(2/sqrt(3))*cos(t), -1/sqrt(3)+(2/sqrt(3))*sin(t),
  t=0..2*Pi],
           [x,0,x=1..12],
           [cos(t),sin(t),t=-Pi/3..Pi/3],
           [Re_gamma_Q(t), Im_gamma_Q(t), t=(-4/3-t_1)..-4/3],
           [Re_gamma_Q(t), Im_gamma_Q(t),
  t=(-4/3-t_2)..(-4/3-t_1)],
           [Re_gamma_Q(t), -Im_gamma_Q(t), t=(-4/3-t_1)..-4/3],
           [Re_gamma_Q(t), -Im_gamma_Q(t),
  t=(-4/3-t_2)..(-4/3-t_1)],
           [cos(t),sin(t),t=Pi/3..5*Pi/3],
           [x,0,x=-5..-1],
```

```
[Re_gamma_Q(t), Im_gamma_Q(t), t=0..t_1],
[Re_gamma_Q(t), Im_gamma_Q(t), t=t_1..t_2],
[Re_gamma_Q(t), -Im_gamma_Q(t), t=0..t_1],
[Re_gamma_Q(t), -Im_gamma_Q(t), t=t_1..t_2]],
color=[yellow, yellow, red, red, blue, blue, blue, blue,
blue,
        green, green, green, green, green],
numpoints=100, scaling=constrained, axes=NONE);
```

$$t\_1 := 5$$

$$t\_2 := 1000$$

$$Invpsi1 := w \rightarrow -1 - \frac{2}{w} - \frac{2\sqrt{1+w}}{w}$$

$$Invpsi1\_plus := w \rightarrow -1 - \frac{2}{w} + \frac{2\sqrt{1+w}}{w}$$

$$Re\_gamma\_Q := t \rightarrow \Re(Invpsi1(Re\_gamma\_P(t) + I\, Im\_gamma\_P(t)))$$

$$Im\_gamma\_Q := t \rightarrow \Im(Invpsi1(Re\_gamma\_P(t) + I\, Im\_gamma\_P(t)))$$



5.E Estimates on Q: Part 1

Lemma 5.16

eps_1, eps_2. [Definition]

```
> eps_1 := 0.057; eps_2 := 0.406;
```

$$eps\_1 := 0.057$$

$$eps\_2 := 0.406$$

(5.2*) This shows that |z - 1|=eps_1 => |Q(z)| < cv*exp(2*Pi*eta).

```
> (2+eps_1)^6/((1-eps_1)*(eps_1)^4); evalf(cv*exp(2*Pi*eta));
```

$$7.610174842\ 10^6$$

$$7.742285500\ 10^6$$

(5.3*) This shows that |z + 1|=eps_2 => |Q(z)| > cv*exp(-2*Pi*eta).

```
> (eps_2)^6/((1+eps_2)*(2+eps_2)^4); evalf(cv*exp(-2*Pi*eta));
```

$$0.00009505793790$$

$$0.00009415824310$$

Parametrize the boundary of E. [Definition]

```
> x := x_E+a_E*t; y := b_E*sqrt(1-t^2);
```

$$x := -0.18 + 1.24\, t$$

$$y := 1.04 \sqrt{1 - t^2}$$

(5.4), (5.7) This shows that D(0,1) is contained in E. [Optional, Exact]

```
> h_1 := x^2 + y^2 -1;
  expand(h_1);
  discrim(h_1, t);
  (0.4464)^2 - 4*0.456*0.114;
```

$$h\_1 := (-0.18 + 1.24\, t)^2 + 0.0816 - 1.0816\, t^2$$

$$0.1140 - 0.4464\, t + 0.4560\, t^2$$

$$-0.008663040000$$

$$-0.00866304$$

(5.5), (5.8) This shows that D(1,eps_1) is contained in E. [Optional, Exact]

```
> h_2 := (x-1)^2 + y^2 - eps_1^2;
  expand(h_2);
  2.9264/(2*0.4560);
  eval(h_2, t=1);
```

$$h\_2 := (-1.18 + 1.24\, t)^2 + 1.078351 - 1.0816\, t^2$$

$$2.470751 - 2.9264\, t + 0.4560\, t^2$$

$$3.208771930$$

$$0.000351$$

(5.6), (5.9) This shows that D(1,eps_2) is contained in E. [Optional, Exact]

```
> h_3 := (x+1)^2+y^2-eps_2^2;
  expand(h_3);
  -2.0336/(2*0.4560);
```

```
eval(h_3, t=-1);
```

$$h\_3 := (0.82 + 1.24\,t)^2 + 0.916764 - 1.0816\,t^2$$

$$1.589164 + 2.0336\,t + 0.4560\,t^2$$

$$-2.229824561$$

$$0.011564$$

The following figure illustrates the statement of Lemma 5.16:

D(0,1) (black), D(1, eps_1) (blue) and D(-1, eps_2) (green) are contained in E (red). [Plot]

```
> plot([[x_E+a_E*cos(t), b_E*sin(t),t=0..2*Pi],
          [cos(t), sin(t),t=0..2*Pi],
          [1+eps_1*cos(t), eps_1*sin(t),t=0..2*Pi],
          [-1+eps_2*cos(t), eps_2*sin(t),t=0..2*Pi]],
          color=[red, black, blue, green], scaling=constrained);
```



Lemma 5.17

eps_3 and eps_4 [Definition]

```
> eps_3 := 2/3; eps_4 := 1.13; r1 := 1.25;
```

$$eps\_3 := \frac{2}{3}$$

$eps\_4 := 1.13$

$r1 := 1.25$

(5.11) This shows that if z in C -(D(i/sqrt(3), 2/sqrt(3)) cup D(-i/sqrt(3), 2/sqrt(3))) and z in D(1, eps_3), then |Q(z)| > R. [Exact]

```
> (4+eps_3^2)^3/((1+eps_3)*eps_3^4); evalf(%); R;
```

$$\frac{800}{3}$$

266.6666667

266

(5.13*) This shows that if z in C -(D(i/sqrt(3), 2/sqrt(3)) cup D(-i/sqrt(3), 2/sqrt(3))) and z in D(-1, eps_4), then |Q(z)| < rho. [Easy]

```
> eps_4^6/(sqrt(1+eps_4^2)*(4+eps_4^2)^2); rho;
```

0.04954969143

0.05

Parametrize Gamma = upper part of boundary of E_1.25. [Definition]

```
> a_Er(r1); b_Er(r1);
  unassin('x','y');
  y := x -> b_Er(1.25)*sqrt(1-((x-x_E)/a_Er(1.25))^2);
```

1.505000000

1.345000000

$unassin(x, y)$

$$y := x \rightarrow b\_Er(1.25)\sqrt{1 - \frac{(x - x\_E)^2}{a\_Er(1.25)^2}}$$

(5.14*) This shows that Gamma_1 is contanied in D(-1, eps_4).

```
> x_E - a_Er(r1); Re_z1 := -1.01;
  x_E - a_Er(r1) + 1;
  h_4 := x -> (x+1)^2 + y(x)^2 - eps_4^2;
  h_4( Re_z1 );
```

-1.685000000

$Re\_z1 := -1.01$

-0.685000000

$$h\_4 := x \rightarrow (1 + x)^2 + y(x)^2 - eps\_4^2$$

-0.017984080

(5.15*), (5.16*) This shows that Gamma_2 is contained in D(i/sqrt(3), 2/sqrt(3)).

```
> Re_z2 := 1.145;
  h_5 := x -> x^2 + (y(x)-1/sqrt(3))^2 - (2/sqrt(3))^2;
  evalf( h_5( Re_z1 ) ); evalf( h_5( Re_z2 ) );
```

$Re\_z2 := 1.145$

$$h\_5 := x \to x^2 + \left(y(x) - \frac{1}{\sqrt{3}} \right)^2 - \frac{4}{\sqrt{3}^2}$$

$$-0.0166230512$$

$$-0.01864870293$$

(5.17*) This shows that Gamma_3 is contained in D(1, eps_3).

```
> x_E + a_Er(r1);
  x_E + a_Er(r1) - 1;
  h_6 := x -> (x-1)^2 + y(x)^2 - eps_3^2;
  h_6( Re_z2 );
```

$$1.325000000$$

$$0.325000000$$

$$h\_6 := x \to (x-1)^2 + y(x)^2 - eps\_3^2$$

$$-0.0165730145$$

The following figure illustrates the statement of Lemma 5.17:
E_1.25 (red) is covered by D(1, eps_3) (blue), D(-1,eps_4) (green), D(i/sqrt(3), 2/sqrt(3)) and D(-i/sqrt(3), 2/sqrt(3)) (magenta) [Plot]

```
> plot([[cos(t), sin(t),t=0..2*Pi],
           [x_E+a_E*cos(t), b_E*sin(t),t=0..2*Pi],
           [1+eps_3*cos(t), eps_3*sin(t),t=0..2*Pi],
           [-1+eps_4*cos(t), eps_4*sin(t),t=0..2*Pi],
           [x_E+a_Er(r1)*cos(t), b_Er(r1)*sin(t),t=0..2*Pi],
           [(2/sqrt(3))*cos(t),
  1/sqrt(3)+(2/sqrt(3))*sin(t),t=0..2*Pi],
           [(2/sqrt(3))*cos(t),
  -1/sqrt(3)+(2/sqrt(3))*sin(t),t=0..2*Pi]],
           color=[gray, pink, blue, green, red, magenta, magenta],
  scaling=constrained);
```



5.F Estimates on Q: Part 2

Lemma 5.19

Q2(z) and its estimate when |z|>r. [Definition]

```
> Q2 := z -> 160/(z-1)^2 + (80*z+32-48/z)/((z-1)^4);
  Q2_max := r -> 160/(r-1)^2 + (80*r+32+48/r)/((r-1)^4);
```

$$Q2 := z \rightarrow \frac{160}{(z-1)^2} + \frac{80\, z + 32 - \frac{48}{z}}{(z-1)^4}$$

$$Q2\_max := r \rightarrow \frac{160}{(r-1)^2} + \frac{80\, r + 32 + \frac{48}{r}}{(r-1)^4}$$

Check that Q(z) = z + 10 + 49/z + Q2(z). [Formal Computation]

```
> simplify(Q(z) - (z+10+49/z+Q2(z)));
```

$$0$$

Lemma 5.20

Evaluate Q2_max(21). [Exact]

```
> Q2_max(21); evalf(%);
```

$$\frac{23}{56}$$

$$0.4107142857$$

Lemma 5.21

DQ(z) = Q'(z). [Definition]

```
> DQ := D(Q);
```

$$DQ := z \to \frac{\left(1+\frac{1}{z}\right)^6}{\left(1-\frac{1}{z}\right)^4} - \frac{6\left(1+\frac{1}{z}\right)^5}{z\left(1-\frac{1}{z}\right)^4} - \frac{4\left(1+\frac{1}{z}\right)^6}{z\left(1-\frac{1}{z}\right)^5}$$

Factor Q'(z). [Formal]

```
> factor(DQ(z));
  factor(DQ(z), sqrt(6));
  simplify( (1-10/z+1/z^2)*(1+1/z)^5/(1-1/z)^5 - DQ(z) );
  simplify( (1-cp/z)*(1-cp2/z) - (1-10/z+1/z^2) );
```

$$\frac{(z+1)^5 (z^2+1-10\,z)}{z^2 (z-1)^5}$$

$$\frac{(z-5-2\sqrt{6})(z-5+2\sqrt{6})(z+1)^5}{z^2 (z-1)^5}$$

$$0$$

$$0$$

Compute the expansion of log(Q'(z)) in two ways. [Formal Computation]

```
> series( log(DQ(z)), z=infinity );
  cp+cp2+5*(-1)-5*1;
  simplify( (cp^2+cp2^2+5*(-1)^2-5*(1)^2)/2 );
  simplify( (cp^3+cp2^3+5*(-1)^3-5*(1)^3)/3 );
  simplify( (cp^4+cp2^4+5*(-1)^4-5*(1)^4)/4 );
```

$$-\frac{49}{z^2} - \frac{320}{z^3} - \frac{4801}{2\,z^4} - \frac{19008}{z^5} + O\left(\frac{1}{z^6}\right)$$

$$0$$

$$49$$

$$320$$

$$\frac{4801}{2}$$

Estimate of | log(Q'(z)) | in |z|>r. [Definition]

```
> LogDQ_max := r -> 49/r^2 + 320/r^3 + (1/4)*( (cp/r)^4/(1-cp/r) +
  (cp2/r)^4/(1-cp2/r) ) + (2/r^5)/(1-1/r^2);
```

$$LogDQ\_max := r \to \frac{49}{r^2} + \frac{320}{r^3} + \frac{cp^4}{4\,r^4\left(1 - \frac{cp}{r}\right)} + \frac{cp2^4}{4\,r^4\left(1 - \frac{cp2}{r}\right)} + \frac{2}{r^5\left(1 - \frac{1}{r^2}\right)}$$

This shows 10*arcsin(1/cp) < 10*(pi/3)*(1/cp) < pi/2. [Optional, Easy]

```
> evalf(10*arcsin(1/cp)); evalf(10*(Pi/3)*(1/cp)); evalf(Pi/2);
```

$$1.011931298$$

$$1.057884353$$

$$1.570796327$$

5.G Estimates on phi

Lemma 5.22

Estimates on phi(z) = z + (c00+c01) + phi1(z).

| c01 | < c01_max, | phi1(z) | < phi1_max(r),

| log(phi'(z)) | < LogDphi_max(r) for |z|>r>1.42. [Definition]

```
> c00 := -x_E;
  c01_max := 2*e_1;
  phi1_max := r -> a_E*sqrt(-log(1-(a_E/(r+x_E))^2));
  LogDphi_max := r -> -log(1-(a_E/(r+x_E))^2);
```

$$c00 := 0.18$$

$$c01\_max := 2.28$$

$$phi1\_max := r \to a\_E\sqrt{-\log\left(1 - \frac{a\_E^2}{(r + x\_E)^2}\right)}$$

$$LogDphi\_max := r \to -\log\left(1 - \frac{a\_E^2}{(r + x\_E)^2}\right)$$

Lemma 5.23

This shows that e_1*r1*(1-1/r1)^2 = 0.057 > rho = 0.05.

```
> r1; e_1*r1*(1-1/r1)^2;
```

$$1.25$$

$$0.05700000000$$

(5.20*) This shows that log((1.25+1)/(1.25-1)) = 2log(3) < 0.7*pi.

```
> (5/4+1)/(5/4-1); log((5/4+1)/(5/4-1)); evalf(%);
  evalf(0.7*Pi);
```

$$9$$

$$2\ln(3)$$

$$2.197224578$$

$$2.199114858$$

Lemma 5.24

r2, r3, theta2, theta3. [Definition]

```
> r2 := 1.4; r3 := 1.54; theta2 := Pi/4; theta3 := 0.4*Pi;
```

$$r2 := 1.4$$

$$r3 := 1.54$$

$$\theta 2 := \frac{1}{4}\pi$$

$$\theta 3 := 0.4\,\pi$$

(5.22*), (5.23*) Verify (5.21) in cases (b) and (c).

```
> log((r3+1)/(r3-1)); evalf(Pi/2);
  (14/10+1)/(14/10-1); log((r2+1)/(r2-1)); evalf(0.6*Pi);
```

$$1.548350221$$

$$1.570796327$$

$$6$$

$$1.791759469$$

$$1.884955592$$

(5.24*) zeta(r2*exp(i*theta2)) is in D(i/sqrt(3),2/sqrt(3)).

```
> (e_1*r2*cos(theta2) + x_E + e_minus1*cos(theta2)/r2)^2
            + (e_1*r2*sin(theta2) - e_minus1*sin(theta2)/r2 -
  1/sqrt(3))^2;
  evalf(%);
```

$$(0.8337142857\sqrt{2} - 0.18)^2 + \left(0.7622857143\sqrt{2} - \frac{1}{3}\sqrt{3}\right)^2$$

$$1.248785997$$

(5.25*) zeta(r3*exp(i*theta3)) is in D(i/sqrt(3),2/sqrt(3)).

```
> (e_1*r3*cos(theta3) + x_E + e_minus1*cos(theta3)/r3)^2
            + (e_1*r3*sin(theta3) - e_minus1*sin(theta3)/r3 -
  1/sqrt(3))^2;
  evalf(%);
```

$$(1.820535065\cos(0.4\,\pi) - 0.18)^2 + \left(1.690664935\sin(0.4\,\pi) - \frac{1}{3}\sqrt{3}\right)^2$$

$$1.208434255$$

(5.26*) zeta(r3*i) is in D(i/sqrt(3),2/sqrt(3)).

```
> (x_E)^2 + (e_1*r3 - e_minus1/r3 - 1/sqrt(3))^2;
  evalf(%);
```

$$0.0324 + \left(1.690664935 - \frac{1}{3}\sqrt{3}\right)^2$$

$$1.271869546$$

The following two figures illustrate the statement of Lemma 5.24.
Re zeta=x_E (gray), D(1,eps_3) (blue), D(-1,eps_4) (green),
E_r1 (red), E_r2 (cyan), E_r3 (gold), zeta({arg w=theta2}) (orange), zeta({arg w=theta3}) (coral),
D(i/sqrt(3),2/sqrt(3)) (magenta). [plot]

```
> plot([[cos(t), sin(t),t=0..Pi], [x_E, t, t=0..2],
          [1+eps_3*cos(t), eps_3*sin(t),t=0..Pi],
          [-1+eps_4*cos(t), eps_4*sin(t), t=0..Pi],
          [x_E+a_E*cos(t), b_E*sin(t),t=0..Pi],
          [x_E+a_Er(r1)*cos(t), b_Er(r1)*sin(t),t=0..Pi],
          [x_E+a_Er(r2)*cos(t), b_Er(r2)*sin(t),t=0..Pi],
          [x_E+a_Er(r3)*cos(t), b_Er(r3)*sin(t),t=0..Pi],
          [x_E+a_Er(t)*cos(theta2), b_Er(t)*sin(theta2),
  t=0.1..r2],
          [x_E+a_Er(t)*cos(theta3), b_Er(t)*sin(theta3),
  t=0.1..r3],
          [(2/sqrt(3))*cos(t),
  1/sqrt(3)+(2/sqrt(3))*sin(t),t=-Pi/3..2*Pi/3]],
          color=[gray, gray, blue, green, pink, red, cyan, gold,
  orange, coral, magenta],
          scaling=constrained, numpoints=100,
  view=[-0.3..1.8,0.0..1.8]);
```



Zoom on Z(r1,r2,theta2), Z(r1,r3,theta3):

Re zeta=x_E (gray), E_r1 (red), E_r2 (cyan), E_r3 (gold),
zeta({arg w=theta2}) (orange), zeta({arg w=theta3}) (coral),, D(i/sqrt(3),2/sqrt(3)) (magenta). [plot].

```
> plot([[x_E, t, t=0.8..1.8],
          [x_E+a_Er(1)*cos(t), b_Er(1)*sin(t),t=0..Pi/2],
          [x_E+a_Er(r1)*cos(t), b_Er(r1)*sin(t),t=0..Pi/2],
          [x_E+a_Er(r2)*cos(t), b_Er(r2)*sin(t),t=0.25*Pi..Pi/2],
          [x_E+a_Er(r3)*cos(t), b_Er(r3)*sin(t),t=0.4*Pi..Pi/2],
          [x_E+a_Er(t)*cos(theta2), b_Er(t)*sin(theta2), t=1..r2],
          [x_E+a_Er(t)*cos(theta3), b_Er(t)*sin(theta3), t=1..r3],
          [(2/sqrt(3))*cos(t),
  1/sqrt(3)+(2/sqrt(3))*sin(t),t=0..2*Pi/3]],
          color=[gray, pink, red, cyan, gold, orange, coral,
  magenta],
          scaling=constrained, numpoints=100,
  view=[-0.2..1.2,0.8..1.8]);
```



5.H Lifting Q and phi to X

Proof of Proposition 5.4 (b)

(5.27*) (5.28) This shows that |phi(zeta)-zeta| < 2.688 if |zeta|>7, and |phi(zeta)| < 9.688 if |zeta| < 7.

```
> evalf( c00 + c01_max + phi1_max(7) );
  evalf( 7 + c00 + c01_max + phi1_max(7) );
```

$$2.687351931$$

$$9.687351931$$

5.I Estimates on f

Lemma 5.27

(a) Estimate the radius of the Euclidean disk which contains f(z)-z (beta_max). [Definition]

```
> beta_max := r -> c01_max + 49/(2*r) + Q2_max(r) + phi1_max(r);
```

$$beta\_max := r \rightarrow c01\_max + \frac{49}{2\,r} + Q2\_max(r) + phi1\_max(r)$$

(b) Estimate |arg (f(z)-z)| for z=phi(zeta) with Re(zeta exp(-i*theta))>r. [Definition]

```
> ArgDeltaF_max := (r, theta) -> -arctan(
  (49*sin(theta)/(2*r))/(10-c00+49*cos(theta)/(2*r)) )
           + arcsin( beta_max(r)/sqrt( (10-c00)^2 + (49/(2*r))^2 +
  2*(10-c00)*(49/(2*r))*cos(theta) ) );
  ArgDeltaF_min := (r, theta) -> -arctan(
  (49*sin(theta)/(2*r))/(10-c00+49*cos(theta)/(2*r)) )
           - arcsin( beta_max(r)/sqrt( (10-c00)^2 + (49/(2*r))^2 +
  2*(10-c00)*(49/(2*r))*cos(theta) ) );
```

$$ArgDeltaF\_max := (r, theta) \rightarrow \text{-arctan}\left(\frac{49\sin(\theta)}{2\,r\left(10 - c00 + \frac{49\cos(\theta)}{2\,r}\right)}\right) + \arcsin\left(\frac{beta\_max(r)}{\sqrt{(10 - c00)^2 + \frac{2401}{4\,r^2} + \frac{49\,(10 - c00)\cos(\theta)}{r}}}\right)$$

$$ArgDeltaF\_min := (r, theta) \rightarrow \text{-arctan}\left(\frac{49\sin(\theta)}{2\,r\left(10 - c00 + \frac{49\cos(\theta)}{2\,r}\right)}\right) - \arcsin\left(\frac{beta\_max(r)}{\sqrt{(10 - c00)^2 + \frac{2401}{4\,r^2} + \frac{49\,(10 - c00)\cos(\theta)}{r}}}\right)$$

(c) Estimate |f(z)-z| for z=phi(zeta) with Re(zeta exp(-i*theta))>r. [Definition]

```
> AbsDeltaF_max := (r, theta) -> sqrt( (10-c00)^2 + (49/(2*r))^2 +
  2*(10-c00)*(49/(2*r))*cos(theta) )
           + beta_max(r);
  AbsDeltaF_min := (r, theta) -> sqrt( (10-c00)^2 + (49/(2*r))^2 +
  2*(10-c00)*(49/(2*r))*cos(theta) )
           - beta_max(r);
```

$$AbsDeltaF\_max := (r, theta) \rightarrow \sqrt{(10 - c00)^2 + \frac{2401}{4\, r^2} + \frac{49\,(10 - c00)\cos(\theta)}{r}} + beta\_max(r)$$

$$AbsDeltaF\_min := (r, theta) \rightarrow \sqrt{(10 - c00)^2 + \frac{2401}{4\, r^2} + \frac{49\,(10 - c00)\cos(\theta)}{r}} - beta\_max(r)$$

(d) Estimate |log f'(z)| for z=phi(zeta) with |zeta|>r. [Definition]

```
> LogDF_max := r -> LogDQ_max(r) + LogDphi_max(r);
```

$$LogDF\_max := r \rightarrow LogDQ\_max(r) + LogDphi\_max(r)$$

(5.29*) This shows that if r>cp, then |alpha|>|beta|.

```
> evalf( 10-c00-49/(2*cp) );
  evalf( beta_max(cp) );
```

7.344997397

7.066419774

5.J Repelling Fatou coordinate tilde Phi_rep on X

Proof of Proposition 5.5

This shows that | arg(10-c0) | < (pi/3)*(2.28/9.82) < pi/6. [Optional, Easy]

```
> evalf( arcsin(c01_max/(10-c00)) );
  evalf( (Pi/3)*c01_max/(10-c00) ); evalf(Pi/10);
```

0.2343175368

0.2431375171

0.3141592654

5.K Attracting Fatou coordinate Phi_attr and domains D_1 and D_1^#

u1, u2, u3, u4

Distance of H_i^+, H_i^- to the origin. [Definition]

```
> u1 := 12.5; u2 := cp; u3 := 27*sqrt(3)/2; u4 := 20.8;
```

$$u1 := 12.5$$

$$u2 := 5 + 2\sqrt{6}$$

$$u3 := \frac{27}{2}\sqrt{3}$$

$$u4 := 20.8$$

Lemma 5.30

(5,30*) and (5.31*) These show that phi(boundary H_2^+) does not intersect H_1^+ and phi(boundary H_4^+) does not intersect H_3^+.

```
> evalf( cp + c00*sqrt(3)/2 + c01_max + phi1_max(cp) );
  evalf( u4 + c00*sqrt(3)/2 + c01_max + phi1_max(u4) );
```

12.49371953

23.31052050

(5.32*) Estimate | arg(f(z)-z)) | for z in H_1^+, hence zeta=phi^{-1}(z) in H_2^+.
This shows that H_1^+ is invariant under f.

```
> evalf( ArgDeltaF_max(cp, Pi/6) ); evalf( -ArgDeltaF_min(cp, Pi/6) );
```

0.5249343444

0.7310821180

Lemma 5.31

r4 and u5, which is the distance between H_3^+ and boundary H_1^+.

```
> r4 := 0.43; u5 := u3 - u1;
```

$$r4 := 0.43$$

$$u5 := \frac{27}{2}\sqrt{3} - 12.5$$

(5.37*) Denote H=H_1+ and suppose z in H_3+, hence zeta=phi^{-1}(z) in H_4^+.
Then proj_+(z) > proj_+(cv)=u3 and u = proj_+(z)-u1 > u5.
Then

f(z) in D_H(z, s(r4)) = D(z + 2*u5*r4^2*exp(i*pi/6)/(1-r4^2), 2*u5*r4/(1-r4^2))

equivalently

f(z)-z = alpha + beta in D(2*u5*r4^2/(1-r4^2), 2*u5*r4/(1-r4^2)),
where alpha = 10 - c00 + 49*exp(-i*pi/6)/(2*u4) and
beta = -c01 + 49*(1/zeta - exp(-i*pi/6)/(2*u4)) + Q2(zeta) - phi1(zeta),

that is

| alpha - 2*u5*r4^2*exp(i*pi/6)/(1-r4^2) | + beta_max(u4) < 2*u5*r4/(1-r4^2).

```
> evalf( sqrt( ( 10 - c00 + 49*sqrt(3)/(4*u4) - sqrt(3)*u5*r4^2/(1-r4^2) )^2
                 + ( 49/(4*u4) + u5*r4^2/(1-r4^2) )^2 )
          + beta_max(u4) - 2*u5*r4/(1-r4^2) );
```

-0.28905990

(5.38*) Estimate arg (Phi_attr'(z)) from above for z in H_3^+ and compare with pi/5.

```
> evalf( -ArgDeltaF_min(u4, Pi/6) + (1/2)*LogDF_max(u4) - (1/2)*log(1-r4^2) );
  evalf( Pi/5 );
```

0.6175222096

0.6283185308

(5.39*) Estimate arg (Phi_attr'(z)) from abobelowve for z in H_3^+ and compare with -pi/6.

```
> evalf( -ArgDeltaF_max(u4, Pi/6) - (1/2)*LogDF_max(u4) + (1/2)*log(1-r4^2) );
  evalf( -Pi/6 );
```

-0.5089687730

-0.5235987758

(5.40*) (5.41*) Estimate | Phi_attr'(z) | in H_3^+ cup H_3^-.

```
> evalf( exp((1/2)*LogDF_max(u4))/(AbsDeltaF_min(u4,
  Pi/6)*sqrt(1-r4^2)) );
  evalf( sqrt(1-r4^2)/(AbsDeltaF_max(u4,
  Pi/6)*exp((1/2)*LogDF_max(u4))) );
```

0.1752613946

0.05580848415

Additional estimates 1: Estimate |f(z)-z| from above and below for z in H_3^+. [Not in the paper]

```
> evalf( AbsDeltaF_max(u4, Pi/6) ); evalf( AbsDeltaF_min(u4, Pi/6)
  );
```

14.80775684

6.904372856

Additional estimates 2: Estimate log (Phi_attr'(z)) from above and below for z in H_3^+. [Not in the paper]

```
> evalf( - log(AbsDeltaF_min(u4, Pi/6)) + (1/2)*LogDF_max(u4) -
  (1/2)*log(1-r4^2) );
  evalf( - log(6.90) + (1/2)*LogDF_max(u4) - (1/2)*log(1-r4^2) );
  evalf( - log(AbsDeltaF_max(u4, Pi/6)) - (1/2)*LogDF_max(u4) +
  (1/2)*log(1-r4^2) );
  evalf( - log(14.81) - (1/2)*LogDF_max(u4) + (1/2)*log(1-r4^2) );
  evalf( exp(-1.74) ); evalf( exp(-2.89) );
  evalf( exp(1.74) ); evalf( exp(2.89) );
```

-1.741476736

-1.740843190

-2.885829376

-2.885980850

0.1755204006

0.05557621261

5.697343423

17.99330960

(5.44*) Estimate that D_1 with eta replaced by eta2 = 13.0 is contained D(0,R1).

```
> R1 := 239; eta2 := 13.0;
  sqrt(1+eta2^2)/0.055;
```

$R1 := 239$

$\eta 2 := 13.0$

237.0619056

5.L Locating domains D_0, D_0', D_{-1} and D_{-1}''

Lemma 5.33

Estimate |Q(zeta)-(zeta+10)| for |zeta|>100. [Optional, Easy]

```
> evalf( 49/100 + Q2_max(100) );
```

0.5064084846

Proof of Lemma 5.32

(5.45) Factorize (Q(z)-cv). [Formal Computation]

```
> factor( Q(z)-cv, sqrt(6) );
```

$$\frac{(z^2 - z + 1)\,(z - 5 + 2\sqrt{6})^2\,(z - 5 - 2\sqrt{6})^2}{z\,(z - 1)^4}$$

This shows that 3*arcsin(1/(cp-2)) < 3*(pi/3)*(1/(cp-2)) < pi/6. [Optional, Easy]

```
> evalf( 3*arcsin(1/(cp-2)) ); evalf( 3*(Pi/3)/(cp-2) ); evalf( Pi/6
  );
```

0.3808177965

0.3977213334

0.5235987758

Q3_max. [Definition]

```
> Q3_max := r -> (80+32/r+48/r^2)/(r^3*(1-1/r)^4);
```

$$Q3\_max := r \to \frac{80 + \frac{32}{r} + \frac{48}{r^2}}{r^3\left(1 - \frac{1}{r}\right)^4}$$

(5.46*) This shows that if zeta is in ell_1^+, then

proj_+(Q(zeta)) < 2*cp/sqrt(3) + 10*sqrt(3)/2 + (49/2)/(2*cp/sqrt(3)) +0 + Q3_max(2*cp/sqrt(3)) < proj_+(cv).

```
> evalf( 2*cp/sqrt(3) + 10*sqrt(3)/2 + (49/2)/(2*cp/sqrt(3)) + 0 +
  Q3_max(2*cp/sqrt(3)) );
  evalf( 27*sqrt(3)/2 );
```

22.31434430

23.38268591

(5.47*) This shows that if zeta is in ell_1^+, then

Im Q(zeta) > cp/sqrt(3) - 49*(3/4)*sqrt(3)/(2*cp) - Q2_max(2cp/sqrt(3)) > 0.

```
> evalf( cp/sqrt(3) - (3*49*sqrt(3))/(8*cp) - Q2_max(2*cp/sqrt(3))
  );
```

0.9490459007

(5.48*) This shows that if Re(zeta) > cp, then Re(Q(zeta)) > 7.6.

```
> evalf( (cp + c00) - (c01_max + phi1_max(cp)) );
```

7.640124015

(5.49*) If proj_+(zeta) > 2*cp/sqrt(3), then proj_+(phi(z)) > 9.1.

Together with the above, this shows that phi(tilde W_0) is conained in W_0 ={Re(z) > 7.6 or proj_+(z) > 9.1 or proj_-(z) > 9.1}.

```
> evalf( (2*cp/sqrt(3) + c00*sqrt(3)/2) - (c01_max +
```

```
phi1_max(2*cp/sqrt(3))) );
```

9.169152471

This shows that if Re(zeta)<0 and 1<|zeta|<7, then |Q(zeta)| < (|zeta|+1/|zeta|) < 7.6. Hence Q(zeta) not in W_0. [Easy]

```
> evalf(7+1/7);
```

7.142857143

If |z|=r > 7 and 2pi/3 < arg z < pi, then |zeta-1|>|zeta|=r>7 and

|zeta Q2(zeta)| < 160/7 + (80*7+32+48/7)/7^3 <160/7 + (80*7+32+143)/7^3 = 25.

[Easy]

```
> 160/7 + (80*7+32+48/7)/7^3; evalf(%); 160/7 + (80*7+32+143)/7^3;
```

$$\frac{59072}{2401}$$

24.60308205

25

```
>
```